第 1 章

Scilab 入門

Scilab (http://www.scilab.org/)[*1]は数値計算を伴う問題の解決手順を記述するのに適したプログラミング言語です．ベクトルや行列を変数の値とすることができ，それらの間の演算が可能なので，パターン認識や機械学習に必要な処理を短いコードで書いて試すことができます．また，データをグラフにして表示する可視化が簡単に行えるため，データ処理の結果や学習結果などを表示させて理解を深めることができます．

ここでは，Scilab の基本的な概念の中から，パターン認識や機械学習のコーディングに用いる側面を選んで説明します．詳細な文法規則や関数の使い方は，下記のオンラインリソースを参照してください．

オンラインヘルプ https://help.scilab.org/
オフィシャルドキュメント http://www.scilab.org/resources/documentation/tutorials

1.1 プログラミング環境

Scilab を起動すると，図 1.1 に示すようなワークスペースが表示されます．中心がコンソールで，ここにコマンドを入力すると，実行結果をすぐに確認することができます．右上のウィンドウは変数ブラウザで，プログラム実行後や中断後の変数の値を見ることができます．

ある程度のまとまった処理は，図 1.2 に示す SciNote をワークスペースから起動してコードを書きます．コーディングの途中で疑問に思った計算式などはコンソールにコピーして実行させてみることができます．

記述したコードは，SciNote ツールバーの右向き三角の実行アイコンをクリックするとコンソールで実行されます．コードに文法的な誤りがあると，コンソールにエラーメッセージが表示されて実行が中断します．その時点までの変数の値が変数ブラウザで確認できるので，多くの場合は原因の同定が容易です．ユーザが定義した関数内でのエラー同定には，ブレークポイントの設定が有効です．ブレークポイントは，コンソールで `setbpt(関数名, 行番号)` と入力して設定し，`delbpt()` で設定を解除します．

1.2 基本

Scilab はインタープリタ型言語で，記述されたコードが上から順に実行されます．プログラミングスタイルとしては，まず定数定義や変数の初期化を行い，次に計算式や関数呼び出しによって必要な計算を

[*1] 本章の説明は ver5.5.2 にもとづいています．



図 1.1　Scilab のワークスペース



図 1.2　SciNote

行って，結果を出力するという手順が基本です．複雑な処理は，ユーザ定義の関数という形でくくり出すことができます．

変数に値を設定する代入文や関数呼び出し文などをコマンドとよびます．コマンドの書き方は以下の規則の通りです．

- コマンドは通常 1 行に 1 つずつ書きます．コンマ (,) でつなげて 1 行に複数のコマンドを書くこともできます．
- コマンドが 1 行に書き切れない場合は，2 つのピリオド (..) を行末に書くと，次の行に継続します．
- コマンドの実行結果はデフォルトで標準出力に表示されます．表示させたくないときは，コマンドの末尾にセミコロン (;) をつけます．
- スラッシュを 2 つ (//) 書くと，その行のそれ以降はコメントになります．

変数

変数には数値・ベクトル・行列・文字列などの型があります．使用前に型を定義する必要はなく，代入時に型が自動的に決まります．変数の使い方は，以下のコード例を参照してください．

`x = 3`	変数 x に整数 3 を代入
`s = 'abc'`	変数 s に文字列 abc を代入
`v = [2, 4, 6]`	変数 v に行ベクトル $(2,4,6)$ を代入
`v(2) = 5`	ベクトル v の第 2 要素を 5 に書き換え ※添字が 1 から始まることに注意
`v = [1, 2, 3]'`	変数 v に列ベクトル $(1,2,3)^T$ を代入
`M = [1, 2, 3; 4, 5, 6]`	変数 M に 2 行 3 列の行列を代入
`[r c] = size(M)`	行列 M の行数 r，列数 c を得る
`M(1,2) = 5`	行列 M の 1 行 2 列目の要素を 5 に書き換え

変数名の命名規則は，先頭文字が非数字（アルファベットの大文字・小文字といくつかの記号）で，その後ろが数字または非数字です．Scilab はあらかじめ用意されている関数が豊富で，数学的な概念を変数名にすると（例えば sum，max，norm など）関数名と衝突して，その関数定義を書き換えてしまうので，注意が必要です．

変数のスコープに関しては，原則的にグローバルで，関数はローカル変数を持つことができるというように理解しておいてください．

また，clear コマンドは，すべての変数を消去します．プログラム実行後もグローバル変数の値は残っているので，デバッグ中など，同じプログラムを何度も実行する場合，変数の値をすべてリセットするために，clear コマンドをプログラムの先頭に書いておくことを推奨します．

基本演算，基本関数

演算子の優先順位は，C 言語などとほぼ同様です．関数の戻り値は，min の例のように複数の値を返すことができる場合もあるので，マニュアルで確認しておいてください．

```
+ 和，- 差，* 積，/ 商，^ べき乗，modulo(m，n) 剰余
abs 絶対値，min 最小値，max 最大値，sqrt 平方，sum 総和
mean 平均，variance 分散，mvvacov 共分散行列，norm ノルム
[a b] = min(v) v の要素の最小値 a，最小値の位置 (argmin) b を得る
```

1.3　行列の扱い

パターン認識では，学習データは特徴ベクトルの集合なので，これを行列として扱うことになります．1 つ分のデータを抜き出す手順や，転置してベクトルとの積を求める手順に慣れておくと，コーディングの速度が速くなります．

基本的な行列の作成

zeros(r,c)	全要素が 0 の r 行 c 列行列
ones(r,c)	全要素が 1 の r 行 c 列行列
eye(r,r)	r 行 r 列の単位行列

部分・範囲の指定

v(m:n)	ベクトル v の第 m 要素から第 n 要素までの部分ベクトル
M(:,n)	行列 M の n 番目の列ベクトル
M(n,:)	行列 M の n 番目の行ベクトル

行列・ベクトルの演算

M’	行列の転置
V1 + V2	ベクトルの和
V1 * V2	ベクトルの積
V1 .* V2	ベクトルの要素毎の積
V1.^2	ベクトルの全要素の 2 乗
cat(dims, M1, M2, ...)	行列の結合．dims=1 は行方向，2 は列方向．

1.4　グラフ表示

ここでは，関数のグラフを表示する方法と 2 次元ベクトルの散布図を表示する方法とを説明します．グラフの表示には plot 関数を使います．関数を表示するためには，その関数をユーザ関数として定義し，表示する範囲を決めて，plot 関数を呼び出します．また，2 次元ベクトルの散布図は横軸の値を第 1 引数，縦軸の値を第 2 引数，点の記号を第 3 引数として plot 関数を呼び出します．

```
function y = f(x)
  y = (1/(sqrt(2))) * exp(-x^2)
endfunction
x=linspace(-2, 2, 50);
plot(x, f)
```




```
X=[1,3,3,7,7,9,10];
Y=[8,7,5,5,4,2,2];
plot(X, Y, "<")
```



1.5　制御

条件分岐を行う if 文は以下のように書きます．条件を記述するための比較演算子は，==，∼=，>=，<=，>，<，論理演算子は　&(and)，|(or)，∼(not) が定義されています．

```
if condition (then)
   body
end
```

繰り返しを行う制御文には for 文と while 文があります．for 文の step は 1 の場合は省略可能です．

```
for variable  =  initial : step : final (do)
   body
end

while condition
   body
end
```

1.6　関数定義

引数 in1，in2，... を受け取り，out の値を返す関数 fname は以下のように定義できます．

```
function out = fname(in1, in2, ...)
  ...
  out = ...
endfunction
```

1.7　知っておくと便利な組み込み関数

特にパターン認識のためのコードを書く際に役に立つ関数をいくつか紹介します．

- mean
  ベクトルや行列の平均を求めます．行列の場合は，平均を計算する単位（'r' は行単位，'c' は列単位）も指定します．
- stdev
  標準偏差を計算します．単位の指定は mean 関数と同じです．この関数は不偏分散から標準偏差を求めています．
- repmat
  ベクトルや行列を複製して，タイル状に配置する関数です．行列に対してベクトルを演算するような場合に威力を発揮します．
- diag

ベクトルの各要素を対角要素とする対角行列を返します.

1.8　演習

- 10 人分のテストの点数を 1 つの変数に格納し，平均点・標準偏差・最高点・最低点などを求めよ.
- 1 人分の身長・体重をまとめたベクトルを 5 人分集めて 1 つの変数に格納し，平均身長・平均体重などを求めよ．また，散布図を表示し，身長と体重の関係を可視化せよ.